
Desain Sistem Kontrol Berbasis Fuzzy Logic Pada Infant Incubator Digital 

ABSTRAK 

Infant incubator digital merupakan alat penunjang kehidupan bayi prematur yang 

memerlukan kestabilan suhu ruang maupun suhu kulit agar proses termoregulasi bayi 

berlangsung optimal. Penelitian ini merancang sistem kontrol suhu pada infant incubator 

digital berbasis Arduino Uno menggunakan metode fuzzy logic Sugeno, dengan dua 

variabel input yaitu error dan delta error suhu, tiga himpunan fungsi keanggotaan segitiga, 

dan sembilan kaidah aturan (rule base). Keluaran fuzzy berupa nilai singleton dikonversi 

menjadi sinyal Time Proportional Control (TPC) untuk mengatur nyala-mati heater melalui 

Solid State Relay serta kecepatan kipas distribusi udara melalui sinyal PWM, berdasarkan 

umpan balik dari sensor suhu ruang SHT35 dan sensor suhu kulit Medke T1307 YSI 400. 

Sistem diuji pada dua mode operasi, yaitu Air mode dengan setpoint 34°C dan Skin mode 

dengan setpoint 37°C, masing-masing selama 30 menit. Hasil pengujian menunjukkan 

bahwa pada Air mode, sistem mencapai kondisi mendekati steady state dalam waktu sekitar 

6 menit dengan steady-state error sebesar 0.2°C (0.59%) dan tanpa overshoot. Pada Skin 

mode, sistem menunjukkan respon yang lebih landai dengan waktu respon lebih dari 30 

menit, namun steady-state error terus mengecil secara konsisten dari 0.9°C menjadi 0.1°C 

tanpa overshoot maupun osilasi berlebih. Hasil ini membuktikan bahwa sistem kontrol 

fuzzy logic Sugeno yang dirancang mampu menjaga kestabilan suhu secara akurat, aman, 

dan adaptif terhadap karakteristik termal yang berbeda pada kedua mode operasi infant 

incubator. 

Kata kunci : Infant Incubator, Fuzzy Logic Sugeno, Arduino Uno, Kontrol Suhu, 

Time Proportional Control. 

 

 

 

 

 

 

 

 

 

 

 

 



 
 

vi 
Politeknik Negeri Jakarta 

Design of a Fuzzy Logic-Based Control System for a Digital Infant Incubator. 

ABSTRACT 

A digital infant incubator is a life-support device for premature infants that requires stable 

ambient and skin temperatures to support optimal thermoregulation. This study designs a 

temperature control system for a digital infant incubator based on Arduino Uno using the 

Sugeno fuzzy logic method, with two input variables, namely temperature error and delta 

error, three triangular membership function sets, and nine rule bases. The fuzzy output, in 

the form of a singleton value, is converted into a Time Proportional Control (TPC) signal 

to regulate the on-off duration of the heater via a Solid State Relay and the speed of the 

distribution fan via a PWM signal, based on feedback from an SHT35 ambient temperature 

sensor and a Medke T1307 YSI 400 skin temperature sensor. The system was tested in two 

operating modes, namely Air mode with a setpoint of 34°C and Skin mode with a setpoint 

of 37°C, each for 30 minutes. The test results show that in Air mode, the system reached a 

near steady-state condition within approximately 6 minutes with a steady-state error of 

0.2°C (0.59%) and no overshoot. In Skin mode, the system exhibited a more gradual 

response with a response time exceeding 30 minutes, yet the steady-state error consistently 

decreased from 0.9°C to 0.1°C without overshoot or excessive oscillation. These results 

demonstrate that the designed Sugeno fuzzy logic control system is able to maintain 

temperature stability accurately, safely, and adaptively across the different thermal 

characteristics of both incubator operating modes. 

Keywords : Infant Incubator, Sugeno Fuzzy Logic, Arduino Uno, Temperature Control, 

Time Proportional Control 

BAB I 

PENDAHULUAN 

1.1 Latar Belakang 

Kemajuan teknologi di industri medis berkembang pesat, terutama dalam 

pengembangan perangkat kesehatan yang memanfaatkan sistem kendali otomatis. 

Infant incubator (inkubator bayi) merupakan contoh perangkat elektro-medis vital 

yang memerlukan sistem kendali canggih untuk menyediakan lingkungan mikro 

yang stabil bagi bayi baru lahir prematur atau Bayi Berat Lahir Rendah (BBLR). 

Bayi prematur belum memiliki kemampuan termoregulasi tubuh yang sempurna 

karena lapisan lemak subkutan yang tipis dan hipotalamus yang belum matang. 

Kondisi ini membuat bayi sangat rentan mengalami hipotermia apabila berada di 

lingkungan dengan fluktuasi suhu yang signifikan. Menurut World Health 

Organization (WHO) dan data Survei Status Gizi Indonesia (2022), tingginya angka 

kelahiran prematur menuntut adanya fasilitas perawatan inkubator yang aman, 

stabil, dan efisien.  

Pada umumnya, sistem inkubator bayi konvensional masih menggunakan 

metode kontrol sederhana seperti ON/OFF. Metode ini memiliki kelemahan berupa 

respon fluktuasi suhu internal yang tidak teratur (Overshoot) karena karakteristik 

pemanasan yang tidak linear, sehingga berisiko memengaruhi kondisi kesehatan 

bayi yang sensitif. Selain itu, banyak inkubator konvensional yang belum 

dilengkapi dengan perangkat antarmuka pemantauan secara real-time, sehingga 

menghambat tenaga medis untuk memonitor kondisi inkubator secara cepat dan 

akurat.  

Untuk mengatasi permasalahan tersebut, diperlukan implementasi 

embedded system menggunakan mikrokontroler andal seperti Arduino Uno yang 

mampu memproses data sensor dan mengendalikan aktuator secara responsif (Sutita 

& Irwan, 2024). Pengaturan kestabilan suhu akan jauh lebih optimal dengan 

menerapkan metode Fuzzy logic sugeno. Metode kontrol cerdas ini bekerja 

menggunakan logika linguistik dan derajat keanggotaan sehingga mampu 

menghasilkan keputusan kontrol yang sangat fleksibel dan presisi dalam menangani 

kondisi lingkungan yang non-linear (Handoyo et al., 2026). Performa sistem kontrol 

ini didukung oleh penggunaan Solid State Relay (SSR) sebagai pengontrol  elemen 

pemanas (heater) karena memiliki kecepatan switching tinggi dan memberikan 

pengendalian beban yang lebih stabil tanpa menghasilkan percikan listrik (Medina 

et al., 2024). (Ravichandran, 2024) menyatakan bahwa sensor SHT35 memiliki 

akurasi dan stabilitas yang tinggi untuk aplikasi suhu ruang. Pada perancangan alat 

ini, sensor tersebut dipadukan dengan sensor Medke T1307 YSI400 untuk 

pemantauan suhu kulit bayi. 

Beberapa perancangan alat terdahulu telah mengimplementasikan metode 

kendali cerdas pada inkubator. Pertama, (Verrel G & Hariyanto, 2022) berhasil 

menstabilkan suhu menggunakan algoritma Fuzzy logic sugeno, namun aktuatornya 

masih mengandalkan saklar mekanik konvensional. Kedua,  (Syaifudin et al., 2026) 

mengembangkan kendali Fuzzy berakurasi tinggi pada prototipe inkubator, tetapi 

belum mengintegrasikan aktuator Solid State Relay (SSR) dan sensor suhu kulit 

bayi. Ketiga, (Co’o et al., 2024) menggunakan Fuzzy Logic dengan pembacaan 

sensor ganda, namun sistemnya difokuskan pada efisiensi baterai untuk inkubator 

portabel (transport), bukan inkubator statis. Berdasarkan ketiga tinjauan tersebut, 

belum ditemukan sistem inkubator statis yang menggabungkan kontrol murni Fuzzy 

logic sugeno, aktuator SSR berbasis time proportional control, pembacaan sensor 

ganda secara bersamaan, dan antarmuka pemantauan real-time berbasis Raspberry 

pi.  

Oleh karena itu, perancangan alat ini dikembangkan untuk mengisi celah 

tersebut melalui Tugas Akhir dengan subjudul “Desain Sistem Kontrol Fuzzy logic 

sugeno Berbasis Arduino Uno pada Infant Incubator Digital”. Sistem ini diharapkan 

mampu menghasilkan pengendalian suhu yang stabil dan akurat, serta 

mempermudah visualisasi data bagi tenaga medis guna mendukung proses 

perawatan bayi prematur secara optimal. 

 

1.2 Rumusan Masalah 

Berdasarkan latar belakang yang telah diuraikan, maka perumusan masalah 

dalam perancangan alat ini adalah sebagai berikut : 

1. Bagaimana cara menerapkan program Fuzzy logic sugeno pada Arduino 

Uno untuk mengatur kerja pemanas (heater) dan kipas (fan) berdasarkan 

data sensor SHT35 dan Medke T1307? 

2. Bagaimana tingkat akurasi dan performa sistem kontrol tersebut dalam 

menjaga kestabilan suhu ruang inkubator setelah dilakukan pengujian? 

 

1.3 Batasan Masalah  

Untuk membatasi ruang lingkup perancangan alat, maka batasan masalah 

pada perancangan alat ini adalah sebagai berikut : 

1. Perancangan alat ini difokuskan pada perancangan dan implementasi sistem 

kontrol suhu pada infant incubator digital menggunakan metode Fuzzy logic 

sugeno berbasis Arduino Uno. 

2. Parameter yang digunakan dalam sistem kontrol ini adalah suhu ruang 

inkubator yang diukur menggunakan sensor SHT35 dan sensor kulit bayi 

yang diukur menggunakan sensor Medke T1307 YSI400. 

3. Pengendalian suhu dilakukan menggunakan elemen pemanas yang 

dikontrol melalui Solid State Relay (SSR), dengan fan DC PWM sebagai 

komponen sirkulasi udara yang kecepatan putarnya diatur oleh sistem fuzzy 

untuk mengoptimalkan distribusi panas di dalam kabin inkubator. 

4. Kelembaban ruang inkubator hanya digunakan sebagai parameter 

monitoring dan tidak dikendalikan secara aktif. 

5. Analisis performa sistem dibatasi pada parameter rise time, Overshoot, 

settling time, response time, dan Steady-state error. 

6. Perancangan alat ini tidak membahas aspek sertifikasi alat kesehatan, uji 

klinis, maupun penggunaan langsung pada pasien atau bayi prematur. 

7. Perancangan alat ini tidak membahas perancangan bentuk fisik inkubator 

secara lengkap, serta sistem antarmuka monitoring berbasis Raspberry pi, 

karena bagian tersebut merupakan ruang lingkup subjudul anggota tim yang 

lain. 

 

1.4 Tujuan 

Tujuan dari Tugas Akhir ini adalah sebagai berikut : 

1. Membuat program algoritma Fuzzy logic sugeno pada mikrokontroler untuk 

mengendalikan sistem pemanasan inkubator secara otomatis.  

2. Mengetahui seberapa akurat dan stabil suhu ruangan serta performa alat 

yang dihasilkan oleh sistem kontrol tersebut melalui hasil pengujian alat. 

 

1.5 Luaran 

Luaran yang diharapkan dari perancangan alat Tugas Akhir ini adalah : 

1. Prototipe sistem kontrol suhu infant incubator berbasis Fuzzy logic sugeno 

yang terintegrasi dengan sensor presisi dan aktuator Solid State Relay (SSR). 

2. Laporan Tugas Akhir  

3. Poster 

4. SOP 

BAB V 

PENUTUP 

 

5.1 Kesimpulan 

Berdasarkan hasil perancangan, pembuatan program, dan pengujian yang telah 

dilakukan pada "Desain Sistem Kontrol Fuzzy logic sugeno Berbasis Arduino Uno 

pada Infant Incubator Digital", maka dapat ditarik kesimpulan sebagai berikut 

untuk menjawab rumusan masalah dan tujuan dari Tugas Akhir ini: 

1. Penerapan Algoritma Fuzzy logic sugeno pada Sistem 

Tujuan untuk membuat program Fuzzy logic sugeno pada mikrokontroler 

telah berhasil direalisasikan untuk mengendalikan suhu secara otomatis. 

Algoritma ini memproses dua variabel input (suhu error dan delta error) 

melalui pemetaan 3 himpunan fungsi keanggotaan segitiga, yang 

menghasilkan 9 kaidah aturan (rule base). Output dari perhitungan fuzzy 

berupa nilai tegas (singleton) yang kemudian dikonversi oleh Arduino Uno 

menjadi sinyal Time Proportional Control (TPC). Sinyal ini secara responsif 

mengatur durasi nyala-mati elemen pemanas (heater) melalui Solid State 

Relay (SSR) pada siklus 5000 ms, sekaligus mengatur kecepatan putar kipas 

distribusi (fan DC) melalui sinyal PWM, berdasarkan data umpan balik dari 

sensor ruang SHT35 dan sensor kulit Medke T1307. 

2. Tingkat Akurasi dan Performa Sistem Kontrol 

Berdasarkan hasil pengujian selama 30 menit pada kedua mode operasi, 

sistem kontrol fuzzy sugeno terbukti mampu menjaga kestabilan suhu tanpa 

menimbulkan overshoot pada Air mode maupun Skin mode. Pada Air mode, 

sistem mencapai kondisi mendekati steady state dengan rise time dan 

settling time sekitar 6 menit serta steady-state error sebesar 0.2°C (0.59%) 

terhadap setpoint 34°C, menunjukkan respon yang cepat dan akurat. Pada 

Skin mode, sistem menunjukkan respon yang lebih landai dengan rise time, 

settling time, dan response time di atas 30 menit karena karakteristik termal 

objek yang diukur lebih lambat, namun steady-state error terus mengecil 

secara konsisten dari 0.9°C menjadi 0.1°C sepanjang pengujian, 

menandakan tren konvergensi menuju setpoint 37°C. Kedua mode sama-

sama tidak menghasilkan peak time maupun overshoot, sehingga dapat 

disimpulkan bahwa sistem kontrol fuzzy sugeno yang dirancang memiliki 

tingkat akurasi dan performa yang baik, aman digunakan, serta mampu 

beradaptasi terhadap karakteristik termal yang berbeda pada masing-masing 

mode operasi infant incubator. 

 

5.2 Saran 

Meskipun prototipe sistem kontrol pada infant incubator digital ini telah bekerja 

dengan baik sesuai perancangan, masih terdapat beberapa potensi pengembangan 

untuk menyempurnakan alat ini di masa mendatang. Beberapa saran yang dapat 

diberikan antara lain: 

1. Penambahan Aktuator Pengendali Kelembaban Udara:  

Pada perancangan alat ini, parameter kelembaban yang dibaca oleh sensor 

SHT35 hanya sebatas ditampilkan untuk kebutuhan monitoring tanpa 

adanya sistem kendali. Untuk pengembangan selanjutnya, disarankan untuk 

menambahkan modul pelembap udara (humidifier atau mist maker) yang 

dikendalikan secara otomatis agar kelembaban kabin dapat dijaga pada 

rentang ideal secara aktif. 

2. Integrasi Sistem Catu Daya Cadangan (UPS):  

Infant incubator merupakan alat penunjang kehidupan (life-support) yang 

tidak boleh mati tiba-tiba saat pemadaman listrik. Pengembangan ke depan 

sangat disarankan untuk menambahkan sistem peralihan catu daya otomatis 

(Uninterruptible Power Supply / UPS) atau baterai cadangan yang 

terintegrasi langsung dengan modul kontrol. 

3. Pengujian Kalibrasi Standar Medis:  

Agar alat dapat dikomersialkan atau digunakan langsung di fasilitas 

kesehatan, disarankan agar prototipe ini diuji kelayakannya menggunakan 

alat kalibrator medis standar seperti INCU Analyzer, untuk memvalidasi 

performa distribusi suhu, tingkat kebisingan, dan kecepatan aliran udara di 

berbagai titik kabin secara lebih komprehensif. 

Lampiran 6 Program Sistem 

/* INKUBATOR BAYI DIGITAL — Arduino Uno 

   Fuzzy Sugeno (2 Input / 2 Output / 9 Rule) 

   Input  : error (setpoint-actual, ±2°C), delta error (±1°C) 

   Output : duty kipas & duty heater (0-255), weighted average defuzzifikasi 

   Protokol serial JSON ke Raspberry Pi, TX tiap 2 detik */ 

 

#include <Wire.h> 

#include <EEPROM.h> 

#include <avr/wdt.h> 

#include <math.h> 

#include <string.h> 

#include <stdlib.h> 

 

// PIN 

#define PIN_THERM      A0 

#define PIN_SSR        9 

#define PIN_FAN        10 

#define PIN_BUZZER     8 

 

#define SHT35_ADDR     0x45 

 

// THERMISTOR (Steinhart-Hart) 

#define THERM_R_SERI   1500.0f 

#define THERM_A        1.412962e-3f 

#define THERM_B        2.514458e-4f 

#define THERM_C        9.235785e-21f 

 

#define SAMPLE_MS      500UL 

#define SEND_MS        2000UL 

#define TPC_WINDOW_MS  5000UL 

#define BAUD_RATE      9600 

 

// SETPOINT 

#define SP_AIR_MIN     32.0f 

#define SP_AIR_MAX     34.0f 

#define SP_AIR_DEF     33.0f 

 

#define SP_SKIN_MIN    35.0f 

#define SP_SKIN_MAX    37.0f 

#define SP_SKIN_DEF    36.0f 

 

// ALARM & PROTEKSI SUHU RUANG 

#define ALARM_WARN     38.0f 

#define ALARM_CRIT     39.0f 

#define TEMP_CUTOFF    39.5f 

 

// ALARM KELEMBABAN (hysteresis) 

#define RH_LOW_ON      40.0f 

#define RH_LOW_OFF     42.0f 

#define RH_HIGH_ON     60.0f 

#define RH_HIGH_OFF    58.0f 

 

// KLASIFIKASI SUHU TUBUH BAYI (mode skin) 

#define BT_HYPO_MAX    36.4f 

#define BT_FEVER_MIN   37.5f 

// SINGLETON OUTPUT — Kipas (0-255) 

#define FAN_VLOW       51    // 20% 

#define FAN_LOW        102   // 40% 

#define FAN_MED        153   // 60% 

#define FAN_HIGH       204   // 80% 

#define FAN_FULL       255   // 100% 

 

// SINGLETON OUTPUT — Heater TPC (0-255) 

#define HEAT_OFF       0     // 0% 

#define HEAT_LOW       64    // 25% 

#define HEAT_MED       128   // 50% 

#define HEAT_HIGH      191   // 75% 

#define HEAT_FULL      255   // 100% 

 

// ANTI-OVERSHOOT 

#define HEAT_CUTOFF_MARGIN   0.2f 

#define FAN_HOLD_BAND        0.3f 

#define FAN_HOLD_DE_BAND     0.15f 

#define FAN_HOLD_LEVEL       FAN_LOW 

 

// EEPROM 

#define EEPROM_ADDR_AIR    0 

#define EEPROM_ADDR_SKIN   4 

#define EEPROM_ADDR_FLAG   8 

#define EEPROM_MAGIC       0xBC 

 

typedef enum : uint8_t { AL_OFF, AL_WARN, AL_CRIT }      AlarmState; 

typedef enum : uint8_t { RH_OK, RH_LOW, RH_HIGH }        RhAlarmState; 

typedef enum : uint8_t { BT_NA, BT_HYPO, BT_NORMAL, BT_FEVER } 

BodyTempState; 

typedef struct { uint8_t fan; uint8_t heat; }             Rule; 

 

// RULE TABLE — baris: Error N/Z/P, kolom: dError N/Z/P 

// Kolom pertama {fan} pakai konstanta FAN_*, kolom kedua {heat} pakai HEAT_*. 

// R1-R4: heater OFF, fan tetap FAN_VLOW untuk sirkulasi (anti panas menumpuk). 

// R5-R9: fan dan heater sama-sama naik proporsional sesuai level error. 

static const Rule ruleTable[3][3] = { 

  { {FAN_VLOW, HEAT_OFF},  {FAN_VLOW, HEAT_OFF},  {FAN_VLOW, 

HEAT_OFF}  }, 

  { {FAN_VLOW, 

HEAT_OFF},  {FAN_LOW,  HEAT_LOW},  {FAN_MED,  HEAT_MED}  }, 

  { {FAN_HIGH, HEAT_HIGH}, {FAN_FULL, HEAT_FULL}, {FAN_FULL, 

HEAT_FULL} } 

}; 

 

// VARIABEL GLOBAL 

char       ctrlMode[5]  = "air"; 

float      spAir        = SP_AIR_DEF; 

float      spSkin       = SP_SKIN_DEF; 

bool       sysRun       = false; 

bool       skinActive   = false; 

bool       coolingMode  = false; 

bool       sendNow      = false; 

float      vSuhu        = 0.0f; 

float      vRh          = 0.0f; 

float      vKulit       = -1.0f; 

 

float      smoothedADC  = 0.0f; 

bool       firstTherm   = true; 

 

uint8_t    outFan       = 0; 

uint8_t    outHeat      = 0; 

 

AlarmState    alarmSt   = AL_OFF; 

RhAlarmState  alarmRh   = RH_OK; 

BodyTempState btSt      = BT_NA; 

 

bool       buzOn        = false; 

unsigned long lastBuzz  = 0UL; 

 

float      prevErr      = 0.0f; 

bool       ssrOn        = false; 

unsigned long lastSample = 0UL; 

unsigned long lastSend   = 0UL; 

unsigned long tpcStart   = 0UL; 

 

void  readSHT35(); 

void  readThermistor(); 

void  runFuzzy(); 

void  updateTPC(); 

void  checkAlarm(); 

void  checkRhAlarm(); 

void  checkBodyTemp(); 

void  updateBuzzer(); 

void  sendData(); 

void  readSerial(); 

void  processCmd(const char* cmd); 

void  sendAck(const char* key, const char* val); 

void  saveEEPROM(); 

void  loadEEPROM(); 

void  setSkinMode(bool active); 

static float triMF(float x, float a, float b, float c); 

 

void setup() { 

  wdt_disable(); 

 

  Serial.begin(BAUD_RATE); 

  Wire.begin(); 

  Wire.setWireTimeout(3000, true); 

 

  pinMode(PIN_SSR,    OUTPUT); 

  pinMode(PIN_FAN,    OUTPUT); 

  pinMode(PIN_BUZZER, OUTPUT); 

  digitalWrite(PIN_SSR,    LOW); 

  digitalWrite(PIN_BUZZER, LOW); 



 

 

69 
Politeknik Negeri Jakarta 

 

  // Timer1: Fast PWM 25 kHz di Pin 10 (OC1B) 

  TCCR1A = (1 << COM1B1) | (1 << WGM11); 

  TCCR1B = (1 << WGM13)  | (1 << WGM12) | (1 << CS10); 

  ICR1   = 639; 

  OCR1B  = 0; 

 

  loadEEPROM(); 

 

  // SHT35 soft-reset 

  Wire.beginTransmission(SHT35_ADDR); 

  Wire.write(0x30); Wire.write(0xA2); 

  Wire.endTransmission(); 

  delay(20); 

 

  unsigned long now = millis(); 

  lastSample = now - SAMPLE_MS; 

  lastSend   = now - SEND_MS; 

  tpcStart   = now; 

 

  wdt_enable(WDTO_8S); 

 

  Serial.println(F("{\"boot\":\"Inkubator v3.9 OK\"}")); 

} 

 

void loop() { 

  wdt_reset(); 

 

  readSerial(); 

  updateTPC(); 

  updateBuzzer(); 

 

  unsigned long now = millis(); 

 

  if ((now - lastSample) >= SAMPLE_MS) { 

    lastSample = now; 

 

    readSHT35(); 

    if (skinActive) readThermistor(); 

 

    checkAlarm(); 

    checkRhAlarm(); 

    checkBodyTemp(); 

    runFuzzy(); 

  } 

 

  if (sendNow || (now - lastSend) >= SEND_MS) { 

    lastSend = now; 

    sendNow  = false; 

    sendData(); 

  } 

} 

// SHT35 — suhu & kelembaban ruang 

void readSHT35() { 

  Wire.beginTransmission(SHT35_ADDR); 

  Wire.clearWireTimeoutFlag(); 

  Wire.write(0x2C); Wire.write(0x06); 

  if (Wire.endTransmission(true) != 0) return; 

  delay(15); 

 

  Wire.requestFrom((uint8_t)SHT35_ADDR, (uint8_t)6); 

  if (Wire.getWireTimeoutFlag()) return; 

  if (Wire.available() < 6) return; 

 

  uint8_t d[6]; 

  for (uint8_t i = 0; i < 6; i++) d[i] = Wire.read(); 

 

  uint16_t rawT = ((uint16_t)d[0] << 8) | d[1]; 

  uint16_t rawH = ((uint16_t)d[3] << 8) | d[4]; 

 

  float t = -45.0f + 175.0f * ((float)rawT / 65535.0f); 

  float h =          100.0f * ((float)rawH / 65535.0f); 

 

  if (t > -10.0f && t < 60.0f)   vSuhu = t; 

  if (h >=  0.0f && h <= 100.0f) vRh   = h; 

} 

 

// Thermistor kulit — hanya Mode Skin 

void readThermistor() { 

  long adcSum = 0; 

  for (int i = 0; i < 10; i++) { adcSum += analogRead(PIN_THERM); delay(2); } 

  float currentADC = (float)adcSum / 10.0f; 

 

  if (currentADC <= 0.0f || currentADC >= 1023.0f) return; 

 

  if (firstTherm) { smoothedADC = currentADC; firstTherm = false; } 

  else             smoothedADC  = 0.1f * currentADC + 0.9f * smoothedADC; 

 

  float R    = THERM_R_SERI * (1023.0f - smoothedADC) / smoothedADC; 

  float lnR  = log(R); 

  float invT = THERM_A + THERM_B * lnR + THERM_C * (lnR * lnR * lnR); 

  float t    = 1.0f / invT - 273.15f; 

 

  if (t > 20.0f && t < 50.0f) vKulit = t; 

} 

 

// Membership function segitiga/shoulder 

static float triMF(float x, float a, float b, float c) { 

  if (a == b) { 

    if (x <= a) return 1.0f; 

    if (x >= c) return 0.0f; 

    return (c - x) / (c - b); 

  } 

  if (b == c) { 

    if (x <= a) return 0.0f; 

    if (x >= b) return 1.0f; 

    return (x - a) / (b - a); 

  } 

  if (x <= a || x >= c) return 0.0f; 

  if (x == b)           return 1.0f; 

  return (x < b) ? ((x - a) / (b - a)) : ((c - x) / (c - b)); 

} 

 

// Fuzzy Sugeno controller 

void runFuzzy() { 

  if (!sysRun) { 

    outFan  = 0; 

    outHeat = 0; 

    OCR1B   = 0; 

    if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

    prevErr = 0.0f; 

    return; 

  } 

 

  if (coolingMode) { 

    outFan  = FAN_FULL; 

    outHeat = HEAT_OFF; 

    OCR1B   = 639; 

    if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

    return; 

  } 

 

  if (vSuhu >= TEMP_CUTOFF) { 

    outFan  = FAN_FULL; 

    outHeat = HEAT_OFF; 

    OCR1B   = 639; 

    if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

    return; 

  } 

 

  float actual, setp; 

  if (skinActive) { 

    actual = vKulit; 

    setp   = spSkin; 

    if (vKulit < 20.0f || vKulit > 50.0f) {  // fail-safe sensor invalid 

      outFan  = FAN_LOW; 

      outHeat = HEAT_OFF; 

      OCR1B   = ((uint32_t)outFan * 639UL) / 255UL; 

      if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

      return; 

    } 

  } else { 

    actual = vSuhu; 

    setp   = spAir; 

    if (vSuhu < 10.0f || vSuhu > 55.0f) {   // fail-safe sensor invalid 

      outFan  = FAN_LOW; 

      outHeat = HEAT_OFF; 

      OCR1B   = ((uint32_t)outFan * 639UL) / 255UL; 

      if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

      return; 

    } 

  } 

 

  float err  = constrain(setp - actual, -2.0f, 2.0f); 

  float dErr = constrain(err - prevErr, -1.0f, 1.0f); 

  prevErr    = err; 

 

  float muE[3]; 

  muE[0] = triMF(err, -2.0f, -2.0f,  0.0f); 

  muE[1] = triMF(err, -1.0f,  0.0f,  1.0f); 

  muE[2] = triMF(err,  0.0f,  2.0f,  2.0f); 

 

  float muDE[3]; 

  muDE[0] = triMF(dErr, -1.0f, -1.0f,  0.0f); 

  muDE[1] = triMF(dErr, -0.5f,  0.0f,  0.5f); 

  muDE[2] = triMF(dErr,  0.0f,  1.0f,  1.0f); 

 

  // REVISI: numFan dan numHeat dihitung independen, masing-masing dari 

  // kolom rule table-nya sendiri (skala FAN_* vs skala HEAT_*), supaya 

  // hasil defuzzifikasi heater akurat sesuai persentase HEAT_* yang benar 

  // (bukan ikut byte skala fan). 

  float numFan = 0.0f, numHeat = 0.0f, den = 0.0f; 

  for (uint8_t i = 0; i < 3; i++) { 

    if (muE[i] < 1e-4f) continue; 

    for (uint8_t j = 0; j < 3; j++) { 

      float mu = (muE[i] < muDE[j]) ? muE[i] : muDE[j]; 

      if (mu < 1e-4f) continue; 

      numFan  += mu * ruleTable[i][j].fan; 

      numHeat += mu * ruleTable[i][j].heat; 

      den     += mu; 

    } 

  } 

 

  if (den > 1e-6f) { 

    outFan  = (uint8_t)constrain((int)(numFan  / den), 0, 255); 

    outHeat = (uint8_t)constrain((int)(numHeat / den), 0, 255); 

  } else { 

    outFan  = FAN_VLOW; 

    outHeat = HEAT_OFF; 

  } 

 

  // Anti-overshoot: predictive heater cutoff (fan tetap jalan, tidak ikut turun) 

  if (err <= HEAT_CUTOFF_MARGIN) { 

    outHeat = HEAT_OFF; 

  } 

 

  // Anti-overshoot: fan hold di sekitar setpoint; heater ikut dikunci 

  // ke level setara (HEAT_LOW), kecuali sudah dipaksa OFF di atas. 

  bool nearSetpoint = (fabsf(err)  <= FAN_HOLD_BAND); 

  bool stableSlope   = (fabsf(dErr) <= FAN_HOLD_DE_BAND); 

  if (nearSetpoint && stableSlope) { 

    outFan = FAN_HOLD_LEVEL; 

    if (err > HEAT_CUTOFF_MARGIN) outHeat = HEAT_LOW; 

  } 

 

  OCR1B = ((uint32_t)outFan * 639UL) / 255UL; 

} 

 

// Time Proportional Control (SSR heater) 

void updateTPC() { 

  unsigned long now  = millis(); 

  unsigned long elap = now - tpcStart; 

  if (elap >= TPC_WINDOW_MS) { tpcStart = now; elap = 0; } 

 

  if (!sysRun || outHeat == 0 || coolingMode) { 

    if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

    return; 

  } 

 

  unsigned long onMs = ((unsigned long)outHeat * TPC_WINDOW_MS) / 255UL; 

  bool want = (elap < onMs); 

  if (want != ssrOn) { 

    digitalWrite(PIN_SSR, want ? HIGH : LOW); 

    ssrOn = want; 

  } 

} 

 

// Alarm & proteksi suhu (hysteresis) 

void checkAlarm() { 

  if (!sysRun) { 

    alarmSt     = AL_OFF; 

    coolingMode = false; 

    return; 

  } 

 

  float val = vSuhu; 

 

  if (val >= ALARM_CRIT) { 

    alarmSt     = AL_CRIT; 

    coolingMode = true; 

    outHeat     = 0; 

    if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

 

  } else if (val >= ALARM_WARN) { 

    alarmSt = AL_WARN; 

 

  } else { 

    alarmSt     = AL_OFF; 

    coolingMode = false; 

  } 

} 

 

// Alarm kelembaban RH, independen dari alarm suhu (hysteresis) 

void checkRhAlarm() { 

  if (!sysRun) { 

    alarmRh = RH_OK; 

    return; 

  } 

 

  switch (alarmRh) { 

    case RH_LOW: 

      if (vRh >= RH_LOW_OFF) alarmRh = RH_OK; 

      break; 

    case RH_HIGH: 

      if (vRh <= RH_HIGH_OFF) alarmRh = RH_OK; 

      break; 

    default: 

      if      (vRh < RH_LOW_ON)  alarmRh = RH_LOW; 

      else if (vRh > RH_HIGH_ON) alarmRh = RH_HIGH; 

      break; 

  } 

} 

 

// Klasifikasi suhu tubuh bayi dari vKulit (khusus Mode Skin) 

void checkBodyTemp() { 

  if (!skinActive || vKulit < 20.0f || vKulit > 50.0f) { 

    btSt = BT_NA; 

    return; 

  } 

 

  if      (vKulit <= BT_HYPO_MAX)  btSt = BT_HYPO; 

  else if (vKulit >= BT_FEVER_MIN) btSt = BT_FEVER; 

  else                              btSt = BT_NORMAL; 

} 

 

// Buzzer — prioritas: alarm suhu CRIT > WARN > alarm RH 

void updateBuzzer() { 

  bool anyAlarm = (alarmSt != AL_OFF) || (alarmRh != RH_OK); 

 

  if (!anyAlarm) { 

    if (buzOn) { digitalWrite(PIN_BUZZER, LOW); buzOn = false; } 

    return; 

  } 

 

  unsigned long now = millis(); 

 

  unsigned long span; 

  if (alarmSt == AL_CRIT) { 

    span = 200UL; 

  } else if (alarmSt == AL_WARN || alarmRh != RH_OK) { 

    span = buzOn ? 200UL : 1000UL; 

  } else { 

    span = 200UL; 

  } 

 

  if ((now - lastBuzz) >= span) { 

    buzOn = !buzOn; 

    digitalWrite(PIN_BUZZER, buzOn ? HIGH : LOW); 

    lastBuzz = now; 

  } 

} 

 

// Kirim status ke Raspberry Pi (JSON, fn/ht dalam persen 0-100) 

void sendData() { 

  Serial.print(F("{\"su\":"));  Serial.print(vSuhu, 1); 

  Serial.print(F(",\"rh\":"));  Serial.print((int)vRh); 

 

  Serial.print(F(",\"sk\":")); 

  if (skinActive && vKulit > 0.0f) Serial.print(vKulit, 1); 

  else                               Serial.print(F("-1")); 

 

  Serial.print(F(",\"fn\":")); Serial.print((int)map(outFan,  0, 255, 0, 100)); 

  Serial.print(F(",\"ht\":")); Serial.print((int)map(outHeat, 0, 255, 0, 100)); 

 

  Serial.print(F(",\"al\":")); 

  if      (alarmSt == AL_CRIT) Serial.print(F("2")); 

  else if (alarmSt == AL_WARN) Serial.print(F("1")); 

  else                          Serial.print(F("0")); 

 

  Serial.print(F(",\"rha\":")); 

  if      (alarmRh == RH_LOW)  Serial.print(F("1")); 

  else if (alarmRh == RH_HIGH) Serial.print(F("2")); 

  else                          Serial.print(F("0")); 

 

  Serial.print(F(",\"bt\":")); 

  if      (btSt == BT_HYPO)   Serial.print(F("1")); 

  else if (btSt == BT_NORMAL) Serial.print(F("2")); 

  else if (btSt == BT_FEVER)  Serial.print(F("3")); 

  else                          Serial.print(F("0")); 

 

  Serial.print(F(",\"md\":")); Serial.print(skinActive ? F("1") : F("0")); 

  Serial.print(F(",\"rn\":")); Serial.print(sysRun ? F("1") : F("0")); 

  Serial.print(F(",\"cl\":")); Serial.print(coolingMode ? F("1") : F("0")); 

 

  Serial.print(F(",\"sa\":")); Serial.print(spAir,  1); 

  Serial.print(F(",\"ss\":")); Serial.print(spSkin, 1); 

  Serial.println(F("}")); 

} 

 

void readSerial() { 

  static char    buf[64]; 

  static uint8_t pos = 0; 

 

  while (Serial.available()) { 

    char c = (char)Serial.read(); 

    if (c == '\n' || c == '\r') { 

      if (pos > 0) { buf[pos] = '\0'; processCmd(buf); } 

      pos = 0; 

    } else if (pos < 63) { 

      buf[pos++] = c; 

    } 

  } 

} 

 

// Perintah dari Raspberry Pi: run, mode, spAir, spSkin 

void processCmd(const char* cmd) { 

  bool  needSave = false; 

  char  ackVal[8]; 

 

  if (strstr(cmd, "\"run\":1")) { 

    if (!sysRun) { 

      sysRun      = true; 

      prevErr     = 0.0f; 

      coolingMode = false; 

      setSkinMode(false); 

 

      tpcStart = millis(); 

      readSHT35(); 

      checkAlarm(); 

      checkRhAlarm(); 

      checkBodyTemp(); 

      runFuzzy(); 

    } 

    sendAck("run", "1"); 

    sendNow = true; 

    return; 

  } 

 

  if (strstr(cmd, "\"run\":0")) { 

    sysRun      = false; 

    coolingMode = false; 

    setSkinMode(false); 

    outFan  = 0; 

    outHeat = 0; 

    OCR1B   = 0; 

    if (ssrOn) { digitalWrite(PIN_SSR, LOW); ssrOn = false; } 

    sendAck("run", "0"); 

    sendNow = true; 

    return; 

  } 

 

  if (strstr(cmd, "\"mode\":\"skin\"")) { 

    if (sysRun) setSkinMode(true); 

    sendAck("mode", "skin"); 

    sendNow = true; 

    return; 

  } 

 

  if (strstr(cmd, "\"mode\":\"air\"")) { 

    setSkinMode(false); 

    sendAck("mode", "air"); 

    sendNow = true; 

    return; 

  } 

 

  const char* p = strstr(cmd, "\"spAir\":"); 

  if (p) { 

    float v = atof(p + 8); 

    v = constrain(v, SP_AIR_MIN, SP_AIR_MAX); 

    if (fabs((double)(v - spAir)) > 0.05f) { spAir = v; needSave = true; } 

    dtostrf(spAir, 4, 1, ackVal); 

    sendAck("spAir", ackVal); 

    if (needSave) saveEEPROM(); 

    sendNow = true; 

    return; 

  } 

 

  p = strstr(cmd, "\"spSkin\":"); 

  if (p) { 

    float v = atof(p + 9); 

    v = constrain(v, SP_SKIN_MIN, SP_SKIN_MAX); 

    if (fabs((double)(v - spSkin)) > 0.05f) { spSkin = v; needSave = true; } 

    dtostrf(spSkin, 4, 1, ackVal); 

    sendAck("spSkin", ackVal); 

    if (needSave) saveEEPROM(); 

    sendNow = true; 

    return; 

  } 

} 

 

void setSkinMode(bool active) { 

  if (active) { 

    strcpy(ctrlMode, "skin"); 

    skinActive = true; 

    firstTherm = true; 

    vKulit     = -1.0f; 

    prevErr    = 0.0f; 

    btSt       = BT_NA; 

  } else { 

    strcpy(ctrlMode, "air"); 

    skinActive = false; 

    vKulit     = -1.0f; 

    prevErr    = 0.0f; 

    btSt       = BT_NA; 

  } 

} 

 

void saveEEPROM() { 

  EEPROM.put(EEPROM_ADDR_AIR,  spAir); 

  EEPROM.put(EEPROM_ADDR_SKIN, spSkin); 

  EEPROM.update(EEPROM_ADDR_FLAG, EEPROM_MAGIC); 

} 

 

void loadEEPROM() { 

  if (EEPROM.read(EEPROM_ADDR_FLAG) != EEPROM_MAGIC) return; 

  EEPROM.get(EEPROM_ADDR_AIR,  spAir); 

  EEPROM.get(EEPROM_ADDR_SKIN, spSkin); 

  spAir  = constrain(spAir,  SP_AIR_MIN,  SP_AIR_MAX); 

  spSkin = constrain(spSkin, SP_SKIN_MIN, SP_SKIN_MAX); 

} 

 

void sendAck(const char* key, const char* val) { 

  Serial.print(F("{\"ack\":\"ok\",\"k\":\"")); 

  Serial.print(key); 

  Serial.print(F("\",\"v\":\"")); 

  Serial.print(val); 

  Serial.println(F("\"}")); 

} 


